


AD-A172 690


ENVIRONMENTAL AND WATER QUALITY OPERATIONAL STUDIES
SIMPLIFIED STEADY-STA. (U) ARMY ENGINEER WATERWAYS
EXPERIMENT STATION VICKSBURG MS ENVIR. J L MARTIN
AUG 86 WES/IR/E-86-4 F/G 8/8


1/1

UNCLASSIFIED

NL







Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

(12)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER Instruction Report E-86-4	2. GOVT ACCESSION NO. AD-A726901	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) SIMPLIFIED, STEADY-STATE TEMPERATURE AND DISSOLVED OXYGEN MODEL: USER'S GUIDE		5. TYPE OF REPORT & PERIOD COVERED Final report
7. AUTHOR(s) James L. Martin		6. PERFORMING ORG. REPORT NUMBER
9. PERFORMING ORGANIZATION NAME AND ADDRESS US Army Engineer Waterways Experiment Station Environmental Laboratory PO Box 631, Vicksburg, Mississippi 39180-0631		8. CONTRACT OR GRANT NUMBER(s)
11. CONTROLLING OFFICE NAME AND ADDRESS DEPARTMENT OF THE ARMY US Army Corps of Engineers Washington, DC 20314-1000		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS EWQOS Work Unit IC3
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)		12. REPORT DATE August 1986
		13. NUMBER OF PAGES 31
		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE
16. DISTRIBUTION STATEMENT (of this Report) Approved for public release; distribution unlimited.		
17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)		
18. SUPPLEMENTARY NOTES Available from National Technical Information Service, 5285 Port Royal Road, Springfield, Virginia 22161.		
19. KEY WORDS (Continue on reverse side if necessary and identify by block number) FORTRAN (Computer program language) Water--Dissolved oxygen Temperature		
20. ABSTRACT (Continue on reverse side if necessary and identify by block number) This report documents the theoretical basis of simplified analytical, steady-state techniques for evaluation of temperature and dissolved oxygen variations in rivers and streams. The derivation and limitations of the analytical techniques are described. A FORTRAN program for application of the techniques is documented and guidance is provided for its use.		

PREFACE

This report was prepared by the Environmental Laboratory (EL) of the US Army Engineer Waterways Experiment Station (WES), as part of the Environmental and Water Quality Operational Studies (EWQOS), Work Unit IC3, "Improve and Verify Riverine Water Quality and Ecological Predictive Techniques." The EWQOS Program was sponsored by the Office, Chief of Engineers (OCE), US Army, and was assigned to WES under the purview of the EL. The OCE Technical Monitors for EWQOS were Mr. Earl Eiker, Dr. John Bushman, and Mr. James L. Gottesman.

This report describes steady-state analytical techniques for the evaluation of water temperatures and dissolved oxygen concentrations in streams and rivers. A FORTRAN program developed for application of these techniques is documented, and guidance for its use is provided.

The study was conducted and the report was prepared by Dr. James L. Martin of the Water Quality Modeling Group (WQMG), Ecosystem Research and Simulation Division (ERSD), EL, under the direct supervision of Mr. Mark S. Dortch, Chief, WQMG, and under the general supervision of Mr. Donald L. Robey, Chief, ERSD, and Dr. John Harrison, Chief, EL. The contributions and reviews of Dr. Marc Zimmerman, Ms. Sandra Bird, and Mr. Dortch, WQMG, are gratefully acknowledged. The report was prepared for publication by Ms. Jessica S. Ruff of the WES Information Products Division.

COL Allen F. Grum, USA, was the previous Director of WES. COL Dwayne G. Lee, CE, is the present Commander and Director. Dr. Robert W. Whalin is the Technical Director.

This report should be cited as follows:

Martin, J. L. 1986. "Simplified, Steady-State Temperature and Dissolved Oxygen Model: User's Guide," Instruction Report E-86-4, US Army Engineer Waterways Experiment Station, Vicksburg, Miss.

CONTENTS

	<u>Page</u>
PREFACE	1
PART I: INTRODUCTION	3
Background	3
Model Development	3
PART II: DATA REQUIREMENTS	12
General	12
Input Data	14
Example Data	18
PART III: OUTPUT DATA DESCRIPTION	20
REFERENCES	25
APPENDIX A: MODEL CODE LISTING	A1

Accession For	
REF ID: A61111	☒
REF ID: A61112	☐
REF ID: A61113	☐

A-1



SIMPLIFIED, STEADY-STATE TEMPERATURE AND DISSOLVED
OXYGEN MODEL: USER'S GUIDE

PART I: INTRODUCTION

Background

1. This report documents the theoretical basis of simplified analytical techniques for estimating water temperature and dissolved oxygen (DO) variations in streams and rivers and provides guidance in their use. These techniques are based upon well-known analytical solutions to mass balance and constituent transport equations. The FORTRAN coding described in this report allows application of these equations to a variety of configurations, including simple river systems, branches, and tributaries. The analytical relationships on which these techniques are based impose limitations that must be considered in the interpretation of results. These assumptions and the details of model development are discussed in the following sections.

2. This model allows comparisons of different flow regimes, inflow loadings, and meteorological conditions on longitudinal spatial distributions of water temperatures and DO concentrations under steady-state conditions. The model has the advantage of ease of application and minimal data requirements and is appropriate where prediction of long-term or time-averaged conditions is suitable for addressing study objectives. The simplicity of the model results from certain assumptions that are detailed in the following section. It is suggested that the user closely review this section to determine if this model is the appropriate tool for his particular needs.

Model Development.

Model assumptions

3. This model is based upon one-dimensional (1-D), longitudinal, steady-state analytical methods for determining streamwise variations

in temperature and DO. This approach is simplified as compared to numerical, time-varying water quality modeling, and offers limited accuracy due to the assumptions on which the model is based.

4. The basic assumptions on which the analytical solutions to the governing equations are based include:

- a. The DO and temperature are uniformly distributed over a given cross section (1-D longitudinal assumption). Thus, only descriptions of longitudinal variations in water quality may be obtained.
- b. Advection dominates the streamflow (dispersive transport is negligible).
- c. All processes are in a steady state (conditions do not change with respect to time).
- d. The DO concentrations are affected only by reaeration and decomposition.

5. The steady-state assumption requires that all flows and boundary conditions be constant for a given simulation. For example, inflow loadings and meteorological data for computing heat exchange are assumed constant. Meteorological data are further assumed to apply to the entire system to be modeled.

6. The analytical equations on which this model is based are applied to a specific river segment to obtain estimates of outflow temperatures and DO concentrations. The depth and water velocity, as determined by the flow rate and channel morphometry, must be constant in a segment (uniform flow). A series of segments in which the flow rate is constant make up a subreach. The entire river system to be modeled, or river reach, may then be composed of a series of subreaches of varying but steady flows (flows do not change with time), as illustrated in Figure 1. Thus, application of the analytical equations to the river system is based upon the following assumptions:

- a. The flow is uniform in each segment, but depths and velocities can vary from segment to segment within a subreach.
- b. The flow rate is constant throughout a subreach but can vary between subreaches.
- c. The streamflow is steady throughout the entire reach.




Figure 1. Example of model segmentation, where segments are represented by rectangles that are numbered consecutively in each subreach. Numbers outside of rectangles indicate subreaches, with the exception of 4, which indicates a lateral inflow

The manner in which segments and subreaches are computationally linked is discussed in the following section.

Temperature analysis

7. Based on the above assumptions, the steady-state uniform flow equation for water temperature over a given segment may be written as

$$Q \frac{dT}{dx} = - \frac{K}{\gamma C_p} \frac{\hat{A}_s A_x}{\hat{V}} (T - T_e) = - \frac{W}{\gamma C_p} (T - T_e) \quad (1)$$

where

Q = streamflow, $m^3 \text{ sec}^{-1}$

T = temperature, $^{\circ}\text{C}$

x = longitudinal distance, m

K = coefficient of surface heat exchange, $J \text{ sec}^{-1} m^{-2} ^{\circ}\text{C}^{-1}$

γ = density of water, $kg \text{ m}^{-3}$

C_p = specific heat of water, $J \text{ kg}^{-1} ^{\circ}\text{C}^{-1}$

$\hat{A}_s$ = segment surface area per unit length, m

A_x = segment cross-sectional area, m^2

$\hat{V}$ = segment volume per unit length, m^2

T_e = equilibrium temperature, $^{\circ}\text{C}$

W = segment top width, $m = \hat{A}_s A_x / V$

8. Equation 1 may be integrated over a uniform segment of length X to

$$T_0 = T_e + (T_i - T_e) \exp \left(- \frac{K A_s}{\gamma C_p Q} X \right) \quad (2)$$

where

T_0 = segment outflow temperature at distance X , $^{\circ}\text{C}$

T_i = segment inflow temperature, $^{\circ}\text{C}$

A_s = surface area, $m^2 = W X$

Considering that $Q = U D W$, where U is the mean velocity for the segment in meters per second and D is hydraulic depth in meters, Equation 2 can be rearranged to yield

$$T_0 = T_e + (T_i - T_e) \exp \left(- \frac{K X}{\gamma C_p D U} \right) \quad (3)$$

9. These formulations allow determination of outflow temperatures for a segment of the river where hydraulic and meteorological data, as well as inflow temperatures, are available. Where temperatures are continuous at the boundaries, the outflow temperature of segment i can serve as the inflow temperature of segment $i+1$, allowing simulation of multiple river segments which are piecewise nonuniform in their flow. This is handled internally in the FORTRAN coding, and the user need only provide the entering temperature for the uppermost segment of subreaches with external boundaries (see Part III: Data Requirements).

10. Provisions are made in the FORTRAN coding of model equations to include application to tributaries, branches, and withdrawals as well as the main branch of the reach. The effects of tributary inflows on water temperatures are computed using a flow-weighted average temperature for the affected subreach's inflow condition.

Dissolved oxygen analysis

11. Assuming that atmospheric reaeration and deoxygenation by decomposition of organic matter are the dominant mechanisms affecting DO concentrations, and given the assumptions described above, the steady-state distributions of biochemical oxygen demand (BOD) and DO may be described by

$$U \frac{dL}{dx} = -K_r L \quad (4)$$

$$U \frac{dC}{dx} = -K_d L + \frac{K_a}{D} (C_s - C) \quad (5)$$

where

$$\begin{aligned} U &= \text{velocity, m sec}^{-1} \\ L &= \text{organic matter (or BOD), g m}^{-3} \\ K_r &= \text{organic matter (BOD) decay rate, sec}^{-1} \\ C &= \text{DO concentration, g m}^{-3} \\ K_d &= \text{deoxygenation rate, sec}^{-1} \\ K_a &= \text{atmospheric exchange rate, m sec}^{-1} \\ C_s &= \text{saturation DO concentration, g m}^{-3} \end{aligned}$$

12. Integrating Equation 4 over the segment length X results in

$$L_0 = L_i \exp(-K_r X/U) \quad (6)$$

where

$$\begin{aligned} L_0 &= \text{outflow BOD concentration, g m}^{-3} \\ L_i &= \text{inflow BOD concentration, g m}^{-3} \end{aligned}$$

Substituting $L(x) = L_i \exp(-K_r X/U)$ for L in Equation 5, where $L(x)$ is the BOD at location x , and integrating over the segment length X results in

$$\begin{aligned} C_0 = C_s - \frac{K_d L_i}{K_a/D - K_r} &\left[\exp\left(-\frac{K_r X}{U}\right) - \exp\left(-\frac{K_a X}{UD}\right) \right] \\ &- (C_s - C_i) \exp\left(-\frac{K_a X}{UD}\right) \end{aligned} \quad (7)$$

where

$$\begin{aligned} C_i &= \text{inflow DO concentration, g m}^{-3} \\ C_0 &= \text{outflow DO concentration, g m}^{-3} \end{aligned}$$

13. If the organic matter, or BOD, is assumed not to vary over the segment, the term $K_d L$ in Equation 5 becomes a constant (equivalent to a zeroth-order approximation) and K_r is zero; the solution to the DO equation reduces to

$$C_O = C_S - \frac{K_O D}{K_a} - \left(C_S - C_i - \frac{K_O D}{K_a} \right) \exp \left(- \frac{K_a X}{UD} \right) \quad (8)$$

where K_O is a zeroth-order oxygen depletion term ($\text{g m}^{-3} \text{sec}^{-1}$).

14. The choice of Equation 7 or 8 for use in simulations may be dictated by the availability of data as well as by actual processes. Where BOD concentrations are unavailable, it may be necessary to estimate a zeroth-order depletion rate. No provisions are made in these formulations to include production or other factors which may influence DO concentrations. Formulations are not included to estimate hydraulic structural reaeration, as from a reregulation dam. This effect can be included by simulating the pool and downstream regions separately, treating them as two different reaches, and incrementing the predicted DO at the structure by an amount reflecting structural reaeration for the inflow condition of the downstream reach. Estimates of hydraulic structural reaeration can be obtained using the empirical relationship developed by Wilhelms and Smith (1981).

15. The model can be applied to tributaries as well as the main branch. The effects of tributary inflows on DO and BOD in the main branch are computed using a flow-weighted average for the affected subreach's inflow conditions.

16. The saturation DO concentration (C_S in Equations 5, 7, and 8) can be estimated from (Mortimer 1981)

$$C_S = P_a \exp \left\{ 7.7117 - 1.31403 [\log_e (T_a + 45.93)] \right\} \quad (9)$$

where T_a is the average temperature for the segment $(T_i + T_o)/2$, in degrees Centigrade, and P_a is an altitude correction coefficient for C_S , computed from

$$P_a = (1.0 - E/44.3)^{5.25} \quad (10)$$

where E is altitude, in kilometers, referenced to mean sea level.

17. The atmospheric exchange coefficient for DO (K_a , in meters per second, Equations 5, 7, and 8) is computed from one of four formulations, with the specific formulation used being dependent upon the needs of the application. These formulations include the expressions of:

Kanwischer (1963)

$$K_a = \frac{D_m}{(200 - 60 W_s^{0.5}) 1.0 \times 10^{-6}} \quad (11)$$

O'Conner and Dobbins (1956) (isotropic turbulence)

$$K_a = 0.00004557 U^{0.5} D^{-0.5} \quad (12)$$

Bennett and Rathbun (1972)

$$K_a = 0.00006215 U^{0.674} D^{-0.865} \quad (13)$$

and Thackston and Krenkel (1969)

$$K_a = 0.0002879 (1.0 + F_r^{0.5}) U^* \quad (14)$$

where

W_s = wind speed, m sec⁻¹

D_m = molecular diffusivity of DO (taken to be
0.204 × 10⁻⁸ m² sec⁻¹)

F_r = Froude number ($U/(gD)^{0.5}$)

U^* = friction velocity, m sec⁻¹ ($C_d W_s^2 R_a / R_w$)^{0.5}

18. Equations 12 and 13 have been multiplied by D to make them consistent with Equations 5, 7, 8, 11, and 14. In the computation of the friction velocity, the drag coefficient (C_d) is taken to be 0.0015. The density of air (R_a) is computed from air temperature (T_{air} , °C) by

$$R_a = 0.00129 - 0.000004 T_{air} \quad (15)$$

and the density of water (R_w) is assumed to be unity.

19. Decomposition is known to be temperature dependent. Additionally, the majority of rate constants are determined at 20° C, requiring that they be adjusted to existing temperature conditions. This is accomplished by

$$K = K_{20} F^{(T - 20)} \quad (16)$$

where

K = rate constant at the temperature T
 K_{20} = rate constant at 20° C
 F = is a constant (taken to be equal to 1.047)

PART II: DATA REQUIREMENTS

General

20. For this simplified model, hydraulic data are not computed internally and must be provided as input data. These data include water velocities and hydraulic depths for each segment (cross section) of given length. The segments should be relatively uniform in their flow. Hydraulic data are required for both the temperature and DO simulations.

21. Meteorological data are required for both the temperature and DO simulations. The temperature model is based upon the equilibrium temperature approach (Edinger, Brady, and Geyer 1974). This approach requires computation of an equilibrium temperature (T_e) and coefficient of heat exchange (K) from representative meteorological data. Wind speed is also required to compute DO reaeration, and air temperatures are required to compute air densities. The computed T_e and K are assumed to apply over the entire reach and are constant for each simulation.

22. The DO concentrations are assumed to vary only due to decomposition and reaeration. Effects of decomposition can be included by using either zeroth- or first-order decay rates. The zeroth-order option requires specification of a single concentration-independent rate of decay. The first-order option requires that an inflow concentration for BOD be provided as well as a decay rate and an oxidation rate. Inflow BOD concentrations should be ultimate BOD. Unless evidence suggests otherwise, the oxidation and decay rates may be equal.

23. Inflow conditions include the entering water temperature and DO concentration for the upstream segment of the main branch or tributary. Inflow BOD concentrations are also required if the first-order process is included.

24. Tributaries, withdrawals, and branches as well as a single main branch can be simulated (Figure 1) where appropriate data are available.

25. A river reach can be broken into a series of consecutively numbered subreaches (eight in Figure 1); that is, all subreaches impacting subreach *i* must be numbered lower than *i*. Each subreach can be subdivided into a number of uniform flow segments for which lengths, depths, and velocities are provided. Subreaches with an external upstream boundary (such as 1 and 2 of Figure 1) must be provided the inflow water temperatures and the DO and BOD (if used) concentrations. Each subreach is also assigned a flow. The flow is used only for book-keeping and for determining inflow concentrations at the junction of subreaches (such as for subreach 3 of Figure 1), and must be reflected in the hydraulic conditions for the subreach. Inflow concentrations in subreaches with tributary flows (such as 3, 5, and 8 of Figure 1) are determined using a flow-weighted average of outflow concentrations from adjoining subreaches. A balance must occur for the specified flows at each junction, or internally computed inflow conditions will be in error. For the case of withdrawals or branches (such as 6 and 7 of Figure 1), inflow concentrations are set equal to the final conditions of the upstream subreach (5, in this example).

26. If it is desired not to carry computations through a tributary or branch as a subreach but to still include its effects on the main branch, zero segments can be specified. For example, flow and inflow temperatures and concentrations should be specified for subreach 4 (Figure 1) but zero segments. Computations would then utilize only flow and specified temperatures and concentrations in the determination of the inflow conditions for subreach 5.

27. The data required as input for simulations and the structure of these data are discussed in the following section. All data provided are in metric units; for convenience, conversion factors are provided in Table 1.

Table 1
Unit Conversion Factors

<u>Multiply</u>	<u>By</u>	<u>To Obtain</u>
BTU ft ⁻² day ⁻¹ °F ⁻¹	0.237	J m ⁻² sec ⁻¹ °C ⁻¹
cubic feet	0.02832	cubic meters
degrees Fahrenheit - 32	0.5556	degrees Centigrade
feet	0.3048	meters
grams per cubic meter	1.000	milligrams per liter
miles (US statute)	1.609	kilometers
miles (US statute) per hour	0.4470	meters per second

Input Data

28. Descriptions of each card image follow. The data are of three types: alphanumeric, real (may be left- or right-justified), and integer (must be right-justified). There are eight data fields per card image:

<u>Field</u>	<u>Length</u>	<u>Columns</u>
1	2	1-2
2	8	3-10
3	8	11-18
4	8	19-26
5	8	27-34
6	8	35-42
7	8	43-50
8	28	51-78

CARD T1

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	T1	-	alphanumeric, Card Identifier
2-8	TITLE1		-	alphanumeric, Descriptive Title

CARD T2

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	T2	-	alphanumeric, Card Identifier
2-8	TITLE2		-	alphanumeric, Descriptive Title

CARD T3

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	T3	-	alphanumeric, Card Identifier
2-8	TITLE3		-	alphanumeric, Descriptive Title

CARD C1

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	C1	-	alphanumeric, Card Identifier
2	IO	0	-	integer, DO not included in simulation
		1	-	integer, DO included in simulations
3	IOPT	0	-	integer, oxygen depletion computed using zeroth-order kinetics
		1	-	integer, oxygen depletion computed using first-order kinetics
4	DECAY	+	day ⁻¹	real, kinetic rate term either zeroth-order (IOPT = 0, K ₀) or first-order (IOPT = 1, K _r) DO depletion

CARD C1 (Cont.)

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
5	OXID	+	day ⁻¹	real kinetic rate term for deoxygenation (K_d , Equation 5). Required if IOPT = 1.
6	NAER	1 to 4	-	Integer, selects reaeration formulation: 1, Kanwischer (1963) 2, O'Conner and Dobbins (1956) 3, Bennett and Rathbun (1972) 4, Thackston and Krenkel (1969)

CARD C2

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	C2	-	alphanumeric, Card Identifier
2	EL	+	m	real, mean elevation (from sea level)
3	TE	+	°C	real, equilibrium temperature
4	CSHE	+	J m ⁻² sec ⁻¹ °C ⁻¹	real, coefficient of surface heat exchange
5	W	+	m sec ⁻¹	real, wind speed
6	TAIR	+	°C	real, air temperature
7	M	+	-	integer, number of subreaches

CARD C3*

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	C3	-	alphanumeric, Card Identifier
2	FLOW	+	m ³ sec ⁻¹	real, flow associated with subreach

* A C3 card is required for each subreach I (I = 1,M).

CARD C3 (Cont.)

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
3	N	+	-	integer, number of segments in subreach
4	NSTART	0	-	integer, indicates an internal upstream boundary for subreach
		1	-	integer, represents an external upstream boundary for subreach
5	NEND	+	-	integer, number of subreach receiving flow
		-1	-	integer, indicates a branch or withdrawal located below subreach
6	N1	+	-	integer, number of downstream branch or withdrawal (required if NEND = -1)
7	N2	+	-	integer, number of second downstream branch or withdrawal (required if NEND = -1)

CARD C4*

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	C4	-	alphanumeric, Card Identifier
2	TO	+	C	real, inflow temperature
3	CO	+	mg/l	real, inflow DO concentration (required if IO = 1)
4	BO	+	mg/l	real, inflow BOD concentration (required if IO = 1 and IOPT = 1)

* A C4 card is required if NSTART = 1 for subreach I (I = 1,M).

CARD DV*

FIELD	PARAMETER	VALUE	UNITS	DESCRIPTION
1	AID	DV	-	alphanumeric, Card Identifier
2	X(I)	+	km	real, segment length
3	U(I)	+	m sec ⁻¹	real, average velocity of segment at given FLOW
4	D(I)	+	m	real, hydraulic depth of segment at given FLOW

* A total of N (I = 1,N) cards should be provided for each subreach.

Example Data

29. Following is an example input data listing for the configuration shown in Figure 1. The output resulting from these data is provided in Part III.

TIEXAMPLE APPLICATION OF STEADY-STATE TEMPERATURE
 T2AND DISSOLVED OXYGEN SIMULATIONS
 T3WES ESRD WQMG J.L. MARTIN AUGUST 1985

C1	1	1	0.15	0.15	4	
C2	540.0	17.8	28.3	5.0	25.1	8
C3	28.3	4	1	3		
C4	9.0	11.0	8.0			
DV	2.25	0.369	1.46			
DV	3.70	0.661	1.02			
DV	3.54	0.180	2.68			
DV	4.83	0.244	1.99			
C3	14.16	4	1	3		
C4	10.3	8.70	3.32			
DV	3.80	0.070	1.54			
DV	2.01	0.116	2.20			
DV	3.15	0.823	0.46			
DV	2.80	0.274	1.02			
C3	42.48	3	0	5		
DV	3.22	0.424	1.29			
DV	3.06	0.302	1.54			
DV	2.90	0.278	1.98			
C3	14.16	0	1	5		
C4	10.3	8.8	1.96			
C3	56.64	2	0	-1	6	7
DV	3.06	0.436	2.03			
DV	3.22	0.567	1.51			
C3	42.48	2	0	8		
DV	3.22	0.424	1.29			
DV	3.06	0.302	1.54			
C3	14.16	2	0	8		
C4	10.3	8.70	3.32			
DV	3.80	0.070	1.54			
C3	56.64	1	0	9		
DV	3.06	0.436	2.03			

PART III: OUTPUT DATA DESCRIPTION

30. Output thermal data are written to TAPE6; DO data are written to TAPE7 (if DO is simulated). Information in these files includes inflow conditions, coefficients and parameters, and results of simulations. The DO concentrations and water temperatures are provided for the downstream boundary of each segment given the upstream boundary conditions. Example output for water temperature using the flow conditions in the example data (paragraph 29) is given on the following pages. (Note that C_s in the DO simulation output refers to saturation DO concentration.)

****STEADY-STATE TEMPERATURE SIMULATIONS ****

**EXAMPLE APPLICATION OF STEADY-STATE TEMPERATURE
AND DISSOLVED OXYGEN SIMULATIONS**

WES ESRD WQMG J.L. MARTIN

AUGUST 1985

SIMULATION CONDITIONS:

COEFFICIENT OF HEAT EXCHANGE= 28.3000 J/M²-SEC-C
EQUILIBRIUM TEMPERATURE= 17.8000 C
WIND SPEED= 5.00 M/SEC

FOR SUBREACH 1 FLOW = 28.3000 (CMS)

ENTERING TEMPERATURE= 9.0000 C

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	TEMPERATURE (C)
2.2500	0.3690	1.4600	9.2448
5.9500	0.6610	1.0200	9.5561
9.4900	0.1800	2.6800	9.9547
14.3200	0.2440	1.9900	10.4645

FOR SUBREACH 2 FLOW = 14.1600 (CMS)

ENTERING TEMPERATURE= 10.3000 C

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	TEMPERATURE (C)
3.8000	0.0700	1.5400	11.8890
5.8100	0.1160	2.2000	12.1952
8.9600	0.8230	0.4600	12.5015
11.7600	0.2740	1.0200	12.8482

FOR SUBREACH 3 FLOW = 42.4800 (CMS)

ENTERING TEMPERATURE= 11.2541 C

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	TEMPERATURE (C)
3.2200	0.4240	1.2900	11.5093
6.2800	0.3020	1.5400	11.7827
9.1800	0.2780	1.9800	11.9931

FOR SUBREACH 4 FLOW = 14.1600 (CMS)

ENTERING TEMPERATURE= 10.3000 C

(Cont inued)

FOR SUBREACH 5 FLOW = 56.6400 (CMS)
ENTERING TEMPERATURE= 11.5698 C

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	TEMPERATURE (C)
3.0600	0.4360	2.0300	11.7136
6.2800	0.5670	1.5100	11.8663

FOR SUBREACH 6 FLOW = 42.4800 (CMS)
ENTERING TEMPERATURE= 11.8663 C

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	TEMPERATURE (C)
3.2200	0.4240	1.2900	12.0976
6.2800	0.3020	1.5400	12.3454

FOR SUBREACH 7 FLOW = 14.1600 (CMS)
ENTERING TEMPERATURE= 11.8663 C

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	TEMPERATURE (C)
10.3000	8.7000	3.3200	11.8806
14.1000	0.0700	1.5400	13.1347

FOR SUBREACH 8 FLOW = 56.6400 (CMS)
ENTERING TEMPERATURE= 12.5428 C

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	TEMPERATURE (C)
3.0600	0.4360	2.0300	12.6641

STEADY-STATE DISSOLVED OXYGEN SIMULATIONS

EXAMPLE APPLICATION OF STEADY-STATE TEMPERATURE
AND DISSOLVED OXYGEN SIMULATIONS

WES ESRD WQMG J.L. MARTIN

AUGUST 1985

SIMULATION CONDITIONS:

DECAY RATE= 0.1500 PER DAY
OXIDATION RATE= 0.1500 PER DAY
WIND SPEED= 5.00 M/S
ELEVATION= 540.00 M

FOR SUBREACH 1 FLOW = 28.3000 (CMS)
ENTERING DO= 11.0000 MG/L
ENTERING BOD= 8.0000 MG/L

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	BOD (MG/L)	Cs (MG/L)	DO (MG/L)
2.2500	0.3690	1.4600	7.9488	10.8082	10.9470
5.9500	0.6610	1.0200	7.9014	10.7369	10.8969
9.4900	0.1800	2.6800	7.7347	10.6470	10.7274
14.3200	0.2440	1.9900	7.5670	10.5340	10.5571

FOR SUBREACH 2 FLOW = 14.1600 (CMS)
ENTERING DO= 8.7000 MG/L
ENTERING BOD= 3.3200 MG/L

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	BOD (MG/L)	Cs (MG/L)	DO (MG/L)
3.8000	0.0700	1.5400	3.1185	10.3197	8.6260
5.8100	0.1160	2.2000	3.0541	10.0986	8.5878
8.9600	0.8230	0.4600	3.0399	10.0289	8.6106
11.7600	0.2740	1.0200	3.0016	9.9556	8.6059

FOR SUBREACH 3 FLOW = 42.4800 (CMS)
ENTERING DO= 9.9018 MG/L
ENTERING BOD= 6.0416 MG/L

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	BOD (MG/L)	Cs (MG/L)	DO (MG/L)
3.2200	0.4240	1.2900	5.9882	10.2518	9.8541
6.2800	0.3020	1.5400	5.9169	10.1900	9.7887
9.1800	0.2780	1.9800	5.8435	10.1340	9.7201

FOR SUBREACH 4 FLOW = 14.1600 (CMS)
ENTERING DO= 8.8000 MG/L
ENTERING BOD= 1.9600 MG/L

(Continued)

FOR SUBREACH 5 FLOW = 56.6400 (CMS)

ENTERING DO= 9.4901 MG/L

ENTERING BOD= 4.8727 MG/L

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	BOD (MG/L)	Cs (MG/L)	DO (MG/L)
3.0600	0.4360	2.0300	4.8324	10.1910	9.4561
6.2800	0.5670	1.5100	4.7998	10.1566	9.4307

FOR SUBREACH 6 FLOW = 42.4800 (CMS).

ENTERING DO= 9.4307 MG/L

ENTERING BOD= 4.7998 MG/L

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	BOD (MG/L)	Cs (MG/L)	DO (MG/L)
3.2200	0.4240	1.2900	4.7562	10.1124	9.3977
6.2800	0.3020	1.5400	4.6980	10.0577	9.3506

FOR SUBREACH 7 FLOW = 14.1600 (CMS)

ENTERING DO= 9.4307 MG/L

ENTERING BOD= 4.7998 MG/L

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	BOD (MG/L)	Cs (MG/L)	DO (MG/L)
10.3000	8.7000	3.3200	4.7930	10.1374	9.4250
14.1000	0.0700	1.5400	4.4833	9.9930	9.1690

FOR SUBREACH 8 FLOW = 56.6400 (CMS)

ENTERING DO= 9.3052 MG/L

ENTERING BOD= 4.6444 MG/L

DISTANCE (KM)	VELOCITY (M/SEC)	DEPTH (M)	BOD (MG/L)	Cs (MG/L)	DO (MG/L)
3.0600	0.4360	2.0300	4.6042	9.9716	9.2710

REFERENCES

- Bennett, J. P., and Rathbun, E. R. 1972. "Reaeration in Open Channel Flow," US Geological Survey Professional Paper 737.
- Edinger, J. E., Brady, D. K., and Geyer, J. C. 1974. "Heat Exchange and Transport in the Environment," Report 14, Electric Power Research Institute, Palo Alto, Calif.
- Kanwischer, J. 1963. "On the Exchange of Gases Between the Atmosphere and the Sea," Deepsea Research with Oceanography, Vol 10, pp 195-207.
- Mortimer, C. H. 1981. "The Oxygen Content of Air Saturated Fresh Waters over Ranges of Temperature and Atmospheric Pressure of Limnological Interest," International Vereinigung for Theoretische und Angewandte Limnologie, Vol 22, pp 2-23.
- O'Conner, D. J., and Dobbins, W. E. 1956. "The Mechanism of Reaeration in Natural Streams," Journal of the Sanitary Engineering Division, ASCE, Vol 82, pp 1-30.
- Thackston, E. L., and Krenkel, P. A. 1969. "Reaeration Prediction in Natural Streams," Journal of the Sanitary Engineering Division, ASCE, Vol 95, pp 65-94.
- Wilhelms, S. C. and Smith, D. R. 1981. "Reaeration Through Gated-Conduit Outlet Works," Technical Report E-81-5, US Army Engineer Waterways Experiment Station, Vicksburg, Miss.

APPENDIX A: MODEL CODE LISTING

```

C *****
C *          STEADY STATE, PIECEWISE UNIFORM FLOW PROGRAM          *
C *          FOR ONE-DIMENSIONAL ANALYSIS OF DO AND              *
C *          TEMPERATURE IN STREAMS AND RIVERS                    *
C *
C *          WATER QUALITY MODELING GROUP                        August 1985 *
C *          WATERWAYS EXPERIMENT STATION                      *
C *          VICKSBURG, MS 39180                                *
C *****

      DIMENSION NSPLIT(20),TOSUM(20),BOSUM(20),COSUM(20)
      CHARACTER*78 TITLE1,TITLE2,TITLE3
C  OPEN INPUT FILE FOR VAX
      OPEN( UNIT=5, FILE='DATA.DAT',STATUS='OLD')
C  OPEN INPUT FILE FOR HARRIS
      CALL ASSIGN(5,7HWOLFDTAT,IERR)

C  READ INPUT DATA
      READ(5,5)TITLE1
      READ(5,5)TITLE2
      READ(5,5)TITLE3
      READ(5,10)IO,IOPT,DECAY,OXID,NAER
      READ(5,15)EL,TE,CSHE,W,TAIR,M
C  OPEN FILES FOR VAX SYSTEM
      IF(IO.EQ.1)OPEN(UNIT=7,FILE='OXY.DAT',STATUS='NEW')
      OPEN(UNIT=6,FILE='TEMP.DAT',STATUS='NEW')
C  OPEN FILES FOR HARRIS
      CALL ASSIGN (7,7HOXYWOLF,IERR)
      CALL ASSIGN (6,7HTEMWOLF,IERR)

C  INITIALIZE COUNTERS

      DO 2 I=1,M
          NSPLIT(I) = 1
          TOSUM(I) = 0.
          BOSUM(I) = 0.
          COSUM(I) = 0.
2      CONTINUE

C  READ FORMATS

      5      FORMAT(2X,A78)
      10     FORMAT(2X,I8,I8,F8.4,F8.4,I8)
      15     FORMAT(2X,F8.4,F8.4,F8.4,F8.4,F8.4,I8)
      20     FORMAT(2X,F8.4,F8.4,F8.4)
      25     FORMAT(2X,F8.4,F8.4,F8.4)
      35     FORMAT(2X,F8.4,I8,I8,I8,I8,I8)
C  VARIABLE DEFINITIONS

```

```

C
C      RHO = WATER DENSITY (KG/M**3)
C      CP  = SPECIFIC HEAT OF WATER (J/KG-C)
C      CS  = SATURATION DISSOLVED OXYGEN CONCENTRATION (G/M**3)
C      D   = HYDRAULIC DEPTH (M)
C      X   = LENGTH OF SEGMENT (M)
C      U   = WATER VELOCITY (M/SEC)
C      CSHE = COEFFICIENT OF SURFACE HEAT EXCHANGE (J/M**2-SEC-C)
C      TE  = EQUILIBRIUM TEMPERATURE (C)
C      T   = WATER TEMPERATURE (C)
C      C   = DISSOLVED OXYGEN CONCENTRATION (G/M**3)
C      CO  = ENTERING DISSOLVED OXYGEN CONCENTRATION (G/M**3)
C      BO  = ENTERING BOD CONCENTRATION
C      ROD = DEPLETION RATE (1/SEC)
C      RBD = BOD DECAY RATE (1/SEC OR MG/L PER SEC)
C      RA  = REAERATION RATE (M/SEC)
C
C
C  DEFINE CONSTANTS
      RHO = 1000.
      CP  = 4190.

C  WRITE HEADER INFORMATION TO OXY.DAT
      IF(IO.EQ.1)THEN
        WRITE(7,11)
11      FORMAT(//10X,'**STEADY-STATE DISSOLVED OXYGEN SIMULATIONS**'//)
        *
        WRITE(7,5)TITLE1
        WRITE(7,5)TITLE2
        WRITE(7,5)TITLE3

        IF(IOPT.NE.1)THEN
          WRITE(7,9)DECAY,W,EL
9        FORMAT(//1X,' SIMULATION CONDITIONS: ',
*          /5X,'DECAY RATE=          ',F8.4,' MG/L PER DAY',
*          /5X,'WIND SPEED=          ',F8.2,' M/SEC',
*          /5X,'ELEVATION=          ',F8.2,' M')
          END IF

          IF(IOPT.EQ.1)THEN
            WRITE(7,12)DECAY,OXID,W,EL
12         FORMAT(//1X,' SIMULATION CONDITIONS: ',
*           /5X,'DECAY RATE=          ',F8.4,' PER DAY',
*           /5X,'OXIDATION RATE=      ',F8.4,' PER DAY',
*           /5X,'WIND SPEED=          ',F8.2,' M/SEC',
*           /5X,'ELEVATION=          ',F8.2,' M')

            END IF

```

```

      END IF

C  WRITE HEADER INFORMATION TO TEMP.DAT

      WRITE(6,21)
21     FORMAT(//10X,'**STEADY-STATE TEMPERATURE SIMULATIONS **//')
      WRITE(6,5)TITLE1
      WRITE(6,5)TITLE2
      WRITE(6,5)TITLE3
      WRITE(6,22)CSHE,TE,W
22     FORMAT(//1X,'SIMULATION CONDITIONS: ',
*        /5X,'COEFFICIENT OF HEAT EXCHANGE= ',F8.4,' J/M**2-SEC-C',
*        /5X,'EQUILIBRIUM TEMPERATURE= ',F8.4,' C',
*        /5X,'WIND SPEED= ',F8.2,' M/SEC')

C  LOOP OVER SUBREACHES

      DO 400 J=1,M

      READ(5,35)FLOW,N,NSTART,NEND,N1,N2

C  SET NSPLIT FOR BRANCHES
      IF(NEND.LT.0)THEN
        NSPLIT(N1)= -1
        NSPLIT(N2)= -1
      END IF

C  SET INITIAL CONDITIONS FOR SUBREACH J

      IF(NSTART.EQ.1)THEN
        READ(5,20)TO,CO,BO
      ELSE
        IF(NSPLIT(J).GT.0)THEN
          TO = TOSUM(J)/FLOW
          CO = COSUM(J)/FLOW
          BO = BOSUM(J)/FLOW
        ELSE
          TO = TOSUM(J)
          CO = COSUM(J)
          BO = BOSUM(J)
        END IF
      END IF

      IF(IO.EQ.1.AND.IOPT.EQ.0)WRITE(7,251)J,FLOW,CO
      IF(IO.EQ.1.AND.IOPT.EQ.1)WRITE(7,252)J,FLOW,CO,BO
      WRITE(6,250)J,FLOW,TO

      IF(N.EQ.0)GOTO 210

C  WRITE HEADER INFORMATION FOR TABLES

```



```

      IF(IOPT.NE.1)WRITE(7,16)
16    FORMAT(////1X,'DISTANCE (KM)  VELOCITY (M/SEC)  DEPTH (M)
      * Cs(MG/L)          DO(MG/L)')
      IF(IOPT.EQ.1)WRITE(7,13)
13    FORMAT(////1X,'DISTANCE (KM)  VELOCITY (M/SEC)  DEPTH (M)
      * BOD(MG/L)        Cs(MG/L)  DO(MG/L)')
      WRITE(6,23)
23    FORMAT(//1X,'DISTANCE (KM)          VELOCITY (M/SEC)          DEPTH (M)
      *   TEMPERATURE (C)')

C  LOOP OVER SEGMENTS IN SUBREACH J
      SUM = 0.
      DO 200 I=1,N

      READ(5,25)X,U,D

      SUM = SUM + X

C  CONVERT SEGMENT LENGTH FROM KM TO M

      X = X * 1000.

C  COMPUTE TEMPERATURE

      R = CSHE*X/(RHO*CP*U*D)
      T = TE + (TO - TE)*EXP(-R)

C  COMPUTE DISSOLVED OXYGEN
      IF(IO.EQ.1) THEN

C      COMPUTE SATURATION DO CONCENTRATION

      TA = (T + TO)/2.
      PALT= (1. - (EL/1000.)/44.3)**5.25
      CS = EXP(7.7117 -1.31403*(LOG(TA+45.93))) *PALT

C      COMPUTE EXCHANGE COEFFICIENT (RA) AND DIVIDE BY DEPTH
C  KANWISCHER (1963)
      IF(NAER.EQ.1)THEN
        DM02 = 0.204E-8
        RA   = (DM02/((200-60.*SQRT(W))*1.E-6))/D
      END IF
C  O'CONNER AND DOBBINS (1956)
      IF(NAER.EQ.2)RA = 0.00004557 * SQRT(U)*D**-1.5
C  BENNETT AND RATHBUN (1972)
      IF(NAER.EQ.3)RA = 0.00006215*U**0.674*D**-1.865
C  THACKSTON AND KRENKEL (1969)
      IF(NAER.EQ.4)THEN
        RHOA = 0.00129-0.000004*TAIR
        FROUDE= U/SQRT(9.8*D)

```

```

      FV      = SQRT(0.0015*RHOA*W*W)
      RA      = 0.0002879*(1.0 + SQRT(FROUDE))*FV/D
      END IF
C  UNIT CONVERSIONS (1/DAY TO 1/SEC) AND TEMPERATURE CORRECTION

      RBD= DECAY/86400. * 1.047**(TA-20.)
      ROD= OXID/86400. * 1.047**(TA-20.)

C      CASE # 1:  ZEROth-ORDER DEPLETION RATE

      IF(IOPT.EQ.0)
*        C  = CS - RBD/RA +(CO-CS+RBD/RA)*EXP(-RA*X/U)

C      CASE # 2:  FIRST-ORDER BOD DECAY RATE

      IF(IOPT.EQ.1)THEN
        R1= EXP(-RA*X/U)
        R2= EXP(-RBD*X/U)
        R3= EXP(-ROD*X/U)

        C  = CS -(ROD*BO/(RA-RBD))*(R2-R1) -(CS-CO)*R1
        B  = BO*R2
      END IF

C  REINITIALIZE INITIAL CONDITIONS FOR NEXT SEGMENT

      CO = C
      BO = B

C  WRITE OUTPUT

      IF(IOPT.EQ.1)WRITE(7,350) SUM,U,D,B,CS,C
      IF(IOPT.NE.1)WRITE(7,350) SUM,U,D,CS,C

      END IF

C  REINITIALIZE INITIAL TEMPERATURE CONDITIONS FOR NEXT SEGMENT

      TO = T

C  WRITE OUTPUT

      WRITE(6,300)SUM,U,D,T

200    CONTINUE

210    WRITE(7,150)
      WRITE(6,150)

C  SUM FLOW-WEIGHTED FINAL CONDITIONS FOR NEW INITIAL CONDITIONS
      IF(NEND.GT.0)THEN

```

```

        COSUM(NEND)= COSUM(NEND) + CO*FLOW
        TOSUM(NEND)= TOSUM(NEND) + TO*FLOW
        BOSUM(NEND)= BOSUM(NEND) + BO*FLOW
    ELSE
        COSUM(N1)= CO
        BOSUM(N1)= BO
        TOSUM(N1)= TO
        COSUM(N2)= CO
        BOSUM(N2)= BO
        TOSUM(N2)= TO
    END IF

400    CONTINUE
C    WRITE FORMAT

150    FORMAT(//1X, '*****')
*****
250    FORMAT(////1X, '  FOR SUBREACH ', I4, ' FLOW = ', F8.4, ' (CMS)'
*      /5X, 'ENTERING TEMPERATURE= ', F8.4, ' C')
251    FORMAT(////1X, '  FOR SUBREACH ', I4, ' FLOW = ', F8.4, ' (CMS)'
*      /5X, 'ENTERING DO= ', F8.4, ' MG/L')
252    FORMAT(////1X, '  FOR SUBREACH ', I4, ' FLOW = ', F8.4, ' (CMS)'
*      /5X, 'ENTERING DO= ', F8.4, ' MG/L',
*      /5X, 'ENTERING BOD= ', F8.4, ' MG/L ')
300    FORMAT(2X, F8.4, 12X, F8.4, 11X, F8.4, 7X, F8.4, 5X, F8.4, 5X, F8.4)
350    FORMAT(2X, F8.4, 7X, F8.4, 6X, F8.4, 4X, F8.4, 5X, F8.4, 3X, F8.4)

    STOP
    END

```

END

1/1-S6

DTIC